Ｊ言語からの群論の理解－その４
２面体(Dihedral)群と置換(Permutation)群
同型(Isomorphism)の意味するもの

西川　利男

０．はじめに－2 面体群(Dihedral Theory)とは

　群論への導入の例として、よく２面体群というのが扱われる。３次元立体の面の角度を２面体角(Dihedral Angle、ちょうど屋根の角度)というのは知っていたが、最初は正３角形や正６角形をなぜ２面体というのか、合点がいかなかった。

　正４面体は４つの正３角形で出来る立体である。正６面体は６つの正方形、正８面体は８つの正３角形、…である。それでは、立体としての裏表を考えた正３角形の板を２面体と呼んでよいのだろう。

　また、ここに群論が始まる、といってもよい。つまり群論とは３次元空間内での立体図形の運動を扱う一種の幾何学の手法とも考えられる。

１．２面体群 D_3（正３角形の板）の対称性[1]

　２面体群 D_3（正３角形の板)の対称操作を考えよう。
２種類の操作がある。
　・三角形板の中心を下から上へと貫く軸の周りの 120° の回転 … r
　・三角形板の１つの頂点からその対辺の中点を結ぶ板の中を通る軸の周りの 180° の回転、つまりこの軸を含む平面の鏡映 … s

　これらから、生成(generate)される６つの操作が２面体群 D_3 の群(Group)を成す。それぞれの操作は群の元(Elements)、元の数は位数(Order)と言う。

$$e,\ r,\ r^2,\ s,\ rs,\ r^2s \qquad (*)$$

　ここで、e は恒等操作（何もしない）。
　　　　　r^2 はｒｒと２度行う。
　　　　　rs は始めにｓを行い、その結果にｒを行う。
　　　　　r^2s は始めにｓを行い、その結果にｒｒと２度行う。
とくに、群操作はＪでいう動詞であり、右から操作する。通常の数学の式演算とは異なることに注意！



[1] M. A. Armstrong, "Groups and Symmetry", p.15-19, Springer(1988).
図は利用しているが、説明は筆者なりに修正してある。

実際に図形の回転操作を行ったようすは次のようになる。



次に、Jのプログラムによりこの操作をシミュレートしてみよう。最初の三角形データをD0とする。なおプログラムのリストは最後にあげた。

```
   D0
  2

3   1
```

これに操作ｓをやってみる。

```
   s D0
  3

2   1
```

この結果に操作ｒを行う。

```
   r s D0
  1

3   2
```

これをまとめてやるには

```
   r@s D0
  1

3   2
```

さらに今の結果に操作ｒを行う。

```
   r@r@s D0
  2

1   3
```

これはｒ ｒ ｓ つまりr^2sと書かれる。

```
   r2s D0
  2

1   3
```

元に戻って、今度は最初の操作をｒとしてやってみる。

```
 r D0
1

2  3
```

そして操作ｓを行う。

```
 s@r D0
2

1  3
```

このように、６つの群操作を適当な順で行えば、すべての対称な図形は得られる。

ここで、操作 sr は群の元(*)のリストにないが、どうするか？上の実験からこれは操作 r^2s に等しい。

また、操作 sr^2 もリストにないがどうするか？実際に図を動かしたり、あるいは上のＪのシミュレーションで行うことができる。

しかし、ここで群の結合の公理(Associative Axiom)に従えば、式の演算のみで結果が得られる。

$$\begin{aligned} sr^2 &= s(rr) = (sr)r = (r^2s)r = r^2(sr) \\ &= r^2(r^2s) = r^4s = r^3(rs) = e(rs) \\ &= rs \end{aligned}$$

群の基本の性質として、個々の操作を結合した操作は必ず群に含まれる別の操作になる。これを示す表は Cayley の群表と呼ばれる。プログラムは稿末に示す。

```
 D3_Table
+---+---+---+---+---+---+---+
|   |e  |r  |r2 |s  |rs |r2s|
+---+---+---+---+---+---+---+
|e  |e  |r  |r2 |s  |rs |r2s|
+---+---+---+---+---+---+---+
|r  |r  |r2 |e  |rs |r2s|s  |
+---+---+---+---+---+---+---+
|r2 |r2 |e  |r  |r2s|s  |rs |
+---+---+---+---+---+---+---+
|s  |s  |r2s|rs |e  |r2 |r  |
+---+---+---+---+---+---+---+
|rs |rs |s  |r2s|r  |e  |r2 |
+---+---+---+---+---+---+---+
|r2s|r2s|rs |s  |r2 |r  |e  |
+---+---+---+---+---+---+---+
```

ところでこれを、先に報告した置換群の場合[2]と比較して検討してみよう。

[2] 西川利男、「J 言語からの群論の理解－その２－置換群」JAPLA 研究会資料 2011/11/26

置換群S_3の置換群操作は次のようになる。

```
 perm_cauchy
+--+-----+
|E |0 1 2|
|  |0 1 2|
+--+-----+
|a |0 1 2|
|  |0 2 1|
+--+-----+
|b |0 1 2|
|  |1 0 2|
+--+-----+
|c |0 1 2|
|  |1 2 0|
+--+-----+
|d |0 1 2|
|  |2 0 1|
+--+-----+
|e |0 1 2|
|  |2 1 0|
+--+-----+
```

これからできる置換群S_3の群表は、次の通りである。

```
 S3_Table
 | E  a  b  c  d  e
-+------------------
E| E  a  b  c  d  e
a| a  E  c  b  e  d
b| b  d  E  e  a  c
c| c  e  a  d  E  b
d| d  b  e  E  c  a
e| e  c  d  a  b  E
```

ここで、置換群S_3の置換操作を２面体群D_3の群操作（回転と鏡映の運動）とつぎのように対応させてみる。

置換群S_3	２面体群D_3
E	e
a	s
b	r^2s
c	r
d	r^2
e	rs

すると群表は置換群S_3も２面体群D_3も全く同じ内容を示していることがわかる。

すなわち、性質の全く異なる２つの操作を、群論では同じように扱うことが出来る。これが同型(Isomorphism)と呼ぶ群論の最大の特徴でありメリットである。

対称性を求めるややこしい立体の運動操作は置換群＝文字の並べ替えで行うことができる。３次元幾何学の対称性を図形に頼らず、式の演算で行えるのである！

先に行ったルービック・キューブの操作も、この文字の並べ替えで行っているにすぎないのである。

```
NB. Dihedral Group D3 / Simulation ====================
D0 =: 3 5$'  2      3   1'
dih =: 3 : 0
:
Z =. y.
Q =. (<0 2){Z
R =. (<2 0){Z
P =. (<2 4){Z
select. x.
  case. 'r' do.
    Z =. R (<2 4) } Z
    Z =. P (<0 2) } Z
    Z =. Q (<2 0) } Z
  case. 's' do.
    Z =. R (<0 2) } Z
    Z =. Q (<2 0) } Z
end.
)

NB. Elements(oper) of D3 group
e   =: ]
r   =: 'r'&dih
r2  =: r@r
s   =: 's'&dih
rs  =: r@s
r2s =: r2@s

gr_exe =: 3 : 0
DX =. y.
select. DX
  case. e   D0 do. 'e'
  case. r   D0 do. 'r '
  case. r2  D0 do. 'r2'
  case. s   D0 do. 's'
  case. rs  D0 do. 'rs'
  case. r2s D0 do. 'r2s'
end.
)

NB. Make Cayley's Group Table
Elem =: 'e';'r';'r2';'s';'rs';'r2s'
TGR0 =: Elem (,/L:0)"(0 1) ' ',L:0 Elem
TGR1 =: 'gr_exe ',L:0 TGR0 (,L:0) ' D0'
TGR =: (". L:0) TGR1
D3_Table =: (7 1$((<' '), Elem)) ,"(1) (1 6$Elem) , TGR
```